

COMP6043 Physical Computing Lab 5

Note: if a task asks you to demonstrate your work to your lecturer, you **must** demonstrate and get your work signed off. Otherwise, no marks will be awarded for your work.

Once completed, upload your report to Canvas.


Pay particular attention when connecting the PCB to the Arduino power pins. It is very important that you connect with the correct polarity, i.e. 5 V on the Arduino to 5 V on the PCB, and GND on the Arduino to GND on the PCB. If you wire up with the incorrect polarity, you will burn out the on-board temperature sensor and damage the board.

This lab is based on the “Digital Hourglass” project in the Arduino Projects Book (p.86). We will replace the tilt switch with a regular push-button switch. When the switch state changes (i.e. ON → OFF, or OFF → ON), the “hourglass” will be reset (i.e. the LEDs will be turned off).

Task 1

[60%]

Wire up your circuit as follows:



Open the Arduino IDE and enter the following code:

```
const int switchPin = 2; // named constant for the switch pin
unsigned long previousTime = 0; // store the last time an LED was updated
int switchState = 0; // the current switch state
int prevSwitchState = 0; // the previous switch state
int led = 3; // a variable to refer to the LEDs
int interval = 5000; // interval at which to light the next LED (5 s)
```

```

void setup() {
    // set the LED pins as outputs and turn LEDs off
    for(int i = 3; i < 6; i++){
        pinMode(i, OUTPUT);
        digitalWrite(i, LOW);
    }
    // set the switch pin as input
    pinMode(switchPin, INPUT);
}

void loop(){
    // store the time since the Arduino started running in a variable
    unsigned long currentTime = millis();

    // compare the current time to the previous time an LED turned on
    // if it is greater than your interval, turn next LED on
    if(currentTime - previousTime > interval && led < 6) {
        // save the current time as the last time you changed an LED
        previousTime = currentTime;
        // Turn the LED on
        digitalWrite(led, HIGH);
        // increment the led variable
        // in 5 s the next LED will light up
        led++;
    }

    // read the switch value
    switchState = digitalRead(switchPin);

    // if the switch has changed
    if(switchState != prevSwitchState){
        // turn all the LEDs low
        for(int i = 3; i < 6; i++){
            digitalWrite(i, LOW);
        }
        // reset the LED variable to the first one
        led = 3;
        //reset the timer
        previousTime = currentTime;
    }

    // set the previous switch state to the current state
    prevSwitchState = switchState;
}

```

Upload your code to the Arduino and verify that your project works correctly.

Demonstrate your project to your lecturer once completed.

Task 2

[10%]

What difference would you observe if you replaced the `loop()` function in the above code with the following? In particular, why is the code from task 1 a better implementation than the below?

```
void loop(){
  if(led < 6) {
    // wait for interval
    delay(interval);
    // Turn the LED on
    digitalWrite(led, HIGH);
    // increment the led variable
    // in 5 s the next LED will light up
    led++;
  }

  // read the switch value
  switchState = digitalRead(switchPin);

  // if the switch has changed
  if(switchState != prevSwitchState){
    // turn all the LEDs low
    for(int i = 3; i < 6; i++){
      digitalWrite(i, LOW);
    }
    // reset the LED variable to the first one
    led = 3;
  }

  // set the previous switch state to the current state
  prevSwitchState = switchState;
}
```

Include your answer in your report.

Task 3

[10%]

The code from task 1 uses an `int` datatype for the `interval` variable but a `long` datatype for `currentTime` and `previousTime`.

- i What would be the effect of declaring `currentTime` and `previousTime` as type `int`?
- ii What change to your code would be required to change the interval from 5 seconds to 10 minutes?

Include your answers in your report.

Task 4

[10%]

Given that `digitalWrite()` sets the voltage on an output pin to 0 V (LOW) or 5 V (HIGH), and that the forward voltage of the LEDs is 1.8 V, what is the current that flows through an LED when:

- i The LED is turned on.
- ii The LED is turned off.

Include your answers in your report.

Task 5

[10%]

Add the S2 switch to your project such that the “hourglass” will only operate if S2 is pressed, i.e. the LEDs should be off if S2 is not pressed.

Note: take care to ensure that the GRN2 LED does not turn on straight away when S2 is pressed, i.e. if the interval is set to 5 s, it should take a minimum of 5 s after S2 is pressed before an LED turns on.

Copy and paste your code into your report.

Demonstrate your project to your lecturer once completed.